

# NAG Library Routine Document

## F08NBF (DGEEVX)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08NBF (DGEEVX) computes the eigenvalues and, optionally, the left and/or right eigenvectors for an  $n$  by  $n$  real nonsymmetric matrix  $A$ .

Optionally, it also computes a balancing transformation to improve the conditioning of the eigenvalues and eigenvectors, reciprocal condition numbers for the eigenvalues, and reciprocal condition numbers for the right eigenvectors.

### 2 Specification

```

SUBROUTINE F08NBF (BALANC, JOBVL, JOBVR, SENSE, N, A, LDA, WR, WI, VL,      &
                  LDVL, VR, LDVR, ILO, IHI, SCALE, ABNRM, RCONDE,      &
                  RCONDV, WORK, LWORK, IWORK, INFO)
INTEGER          N, LDA, LDVL, LDVR, ILO, IHI, LWORK, IWORK(*), INFO
REAL (KIND=nag_wp) A(LDA,*), WR(*), WI(*), VL(LDVL,*), VR(LDVR,*),      &
                  SCALE(*), ABNRM, RCONDE(*), RCONDV(*),              &
                  WORK(max(1,LWORK))
CHARACTER(1)    BALANC, JOBVL, JOBVR, SENSE

```

The routine may be called by its LAPACK name *dgeevx*.

### 3 Description

The right eigenvector  $v_j$  of  $A$  satisfies

$$Av_j = \lambda_j v_j$$

where  $\lambda_j$  is the  $j$ th eigenvalue of  $A$ . The left eigenvector  $u_j$  of  $A$  satisfies

$$u_j^H A = \lambda_j u_j^H$$

where  $u_j^H$  denotes the conjugate transpose of  $u_j$ .

Balancing a matrix means permuting the rows and columns to make it more nearly upper triangular, and applying a diagonal similarity transformation  $DAD^{-1}$ , where  $D$  is a diagonal matrix, with the aim of making its rows and columns closer in norm and the condition numbers of its eigenvalues and eigenvectors smaller. The computed reciprocal condition numbers correspond to the balanced matrix. Permuting rows and columns will not change the condition numbers (in exact arithmetic) but diagonal scaling will. For further explanation of balancing, see Section 4.8.1.2 of Anderson *et al.* (1999).

Following the optional balancing, the matrix  $A$  is first reduced to upper Hessenberg form by means of unitary similarity transformations, and the  $QR$  algorithm is then used to further reduce the matrix to upper triangular Schur form,  $T$ , from which the eigenvalues are computed. Optionally, the eigenvectors of  $T$  are also computed and backtransformed to those of  $A$ .

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

## 5 Parameters

- 1: BALANC – CHARACTER(1) *Input*
- On entry:* indicates how the input matrix should be diagonally scaled and/or permuted to improve the conditioning of its eigenvalues.
- BALANC = 'N'  
Do not diagonally scale or permute.
- BALANC = 'P'  
Perform permutations to make the matrix more nearly upper triangular. Do not diagonally scale.
- BALANC = 'S'  
Diagonally scale the matrix, i.e., replace  $A$  by  $DAD^{-1}$ , where  $D$  is a diagonal matrix chosen to make the rows and columns of  $A$  more equal in norm. Do not permute.
- BALANC = 'B'  
Both diagonally scale and permute  $A$ .
- Computed reciprocal condition numbers will be for the matrix after balancing and/or permuting. Permuting does not change condition numbers (in exact arithmetic), but balancing does.
- Constraint:* BALANC = 'N', 'P', 'S' or 'B'.
- 2: JOBVL – CHARACTER(1) *Input*
- On entry:* if JOBVL = 'N', the left eigenvectors of  $A$  are not computed.  
If JOBVL = 'V', the left eigenvectors of  $A$  are computed.  
If SENSE = 'E' or 'B', JOBVL must be set to JOBVL = 'V'.
- Constraint:* JOBVL = 'N' or 'V'.
- 3: JOBVR – CHARACTER(1) *Input*
- On entry:* if JOBVR = 'N', the right eigenvectors of  $A$  are not computed.  
If JOBVR = 'V', the right eigenvectors of  $A$  are computed.  
If SENSE = 'E' or 'B', JOBVR must be set to JOBVR = 'V'.
- Constraint:* JOBVR = 'N' or 'V'.
- 4: SENSE – CHARACTER(1) *Input*
- On entry:* determines which reciprocal condition numbers are computed.
- SENSE = 'N'  
None are computed.
- SENSE = 'E'  
Computed for eigenvalues only.
- SENSE = 'V'  
Computed for right eigenvectors only.
- SENSE = 'B'  
Computed for eigenvalues and right eigenvectors.
- If SENSE = 'E' or 'B', both left and right eigenvectors must also be computed (JOBVL = 'V' and JOBVR = 'V').
- Constraint:* SENSE = 'N', 'E', 'V' or 'B'.

- 5: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 6: A(LDA,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  matrix  $A$ .  
*On exit:*  $A$  has been overwritten. If  $\text{JOBVL} = 'V'$  or  $\text{JOBVR} = 'V'$ ,  $A$  contains the real Schur form of the balanced version of the input matrix  $A$ .
- 7: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F08NBF (DGEEVX) is called.  
*Constraint:*  $\text{LDA} \geq \max(1, N)$ .
- 8: WR(\*) – REAL (KIND=nag\_wp) array *Output*  
9: WI(\*) – REAL (KIND=nag\_wp) array *Output*  
**Note:** the dimension of the arrays  $\text{WR}$  and  $\text{WI}$  must be at least  $\max(1, N)$ .  
*On exit:*  $\text{WR}$  and  $\text{WI}$  contain the real and imaginary parts, respectively, of the computed eigenvalues. Complex conjugate pairs of eigenvalues appear consecutively with the eigenvalue having the positive imaginary part first.
- 10: VL(LDVL,\*) – REAL (KIND=nag\_wp) array *Output*  
**Note:** the second dimension of the array  $\text{VL}$  must be at least  $\max(1, N)$  if  $\text{JOBVL} = 'V'$ , and at least 1 otherwise.  
*On exit:* if  $\text{JOBVL} = 'V'$ , the left eigenvectors  $u_j$  are stored one after another in the columns of  $\text{VL}$ , in the same order as their corresponding eigenvalues. If the  $j$ th eigenvalue is real, then  $u_j = \text{VL}(:, j)$ , the  $j$ th column of  $\text{VL}$ . If the  $j$ th and  $(j + 1)$ st eigenvalues form a complex conjugate pair, then  $u_j = \text{VL}(:, j) + i \times \text{VL}(:, j + 1)$  and  $u_{j+1} = \text{VL}(:, j) - i \times \text{VL}(:, j + 1)$ .  
If  $\text{JOBVL} = 'N'$ ,  $\text{VL}$  is not referenced.
- 11: LDVL – INTEGER *Input*  
*On entry:* the first dimension of the array  $\text{VL}$  as declared in the (sub)program from which F08NBF (DGEEVX) is called.  
*Constraints:*  
if  $\text{JOBVL} = 'V'$ ,  $\text{LDVL} \geq \max(1, N)$ ;  
otherwise  $\text{LDVL} \geq 1$ .
- 12: VR(LDVR,\*) – REAL (KIND=nag\_wp) array *Output*  
**Note:** the second dimension of the array  $\text{VR}$  must be at least  $\max(1, N)$  if  $\text{JOBVR} = 'V'$ , and at least 1 otherwise.  
*On exit:* if  $\text{JOBVR} = 'V'$ , the right eigenvectors  $v_j$  are stored one after another in the columns of  $\text{VR}$ , in the same order as their corresponding eigenvalues. If the  $j$ th eigenvalue is real, then  $v_j = \text{VR}(:, j)$ , the  $j$ th column of  $\text{VR}$ . If the  $j$ th and  $(j + 1)$ st eigenvalues form a complex conjugate pair, then  $v_j = \text{VR}(:, j) + i \times \text{VR}(:, j + 1)$  and  $v_{j+1} = \text{VR}(:, j) - i \times \text{VR}(:, j + 1)$ .  
If  $\text{JOBVR} = 'N'$ ,  $\text{VR}$  is not referenced.

- 13: LDVR – INTEGER *Input*  
*On entry:* the first dimension of the array VR as declared in the (sub)program from which F08NBF (DGEEVX) is called.  
*Constraints:*  
     if JOBVR = 'V', LDVR  $\geq$  max(1, N);  
     otherwise LDVR  $\geq$  1.
- 14: ILO – INTEGER *Output*  
 15: IHI – INTEGER *Output*  
*On exit:* ILO and IHI are integer values determined when *A* was balanced. The balanced *A* has  $a_{ij} = 0$  if  $i > j$  and  $j = 1, 2, \dots, ILO - 1$  or  $i = IHI + 1, \dots, N$ .
- 16: SCALE(\*) – REAL (KIND=nag\_wp) array *Output*  
**Note:** the dimension of the array SCALE must be at least max(1, N).  
*On exit:* details of the permutations and scaling factors applied when balancing *A*.  
 If  $p_j$  is the index of the row and column interchanged with row and column  $j$ , and  $d_j$  is the scaling factor applied to row and column  $j$ , then  
     SCALE( $j$ ) =  $p_j$ , for  $j = 1, 2, \dots, ILO - 1$ ;  
     SCALE( $j$ ) =  $d_j$ , for  $j = ILO, \dots, IHI$ ;  
     SCALE( $j$ ) =  $p_j$ , for  $j = IHI + 1, \dots, N$ .  
 The order in which the interchanges are made is N to IHI + 1, then 1 to ILO - 1.
- 17: ABNRM – REAL (KIND=nag\_wp) *Output*  
*On exit:* the 1-norm of the balanced matrix (the maximum of the sum of absolute values of elements of any column).
- 18: RCONDE(\*) – REAL (KIND=nag\_wp) array *Output*  
**Note:** the dimension of the array RCONDE must be at least max(1, N).  
*On exit:* RCONDE( $j$ ) is the reciprocal condition number of the  $j$ th eigenvalue.
- 19: RCONDV(\*) – REAL (KIND=nag\_wp) array *Output*  
**Note:** the dimension of the array RCONDV must be at least max(1, N).  
*On exit:* RCONDV( $j$ ) is the reciprocal condition number of the  $j$ th right eigenvector.
- 20: WORK(max(1, LWORK)) – REAL (KIND=nag\_wp) array *Workspace*  
*On exit:* if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimal performance.
- 21: LWORK – INTEGER *Input*  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08NBF (DGEEVX) is called.  
 If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.  
*Suggested value:* for optimal performance, LWORK must generally be larger than the minimum, increase LWORK by, say,  $N \times nb$ , where  $nb$  is the optimal **block size** for F08NEF (DGEHRD).

*Constraints:*

if JOBVL = 'N' and JOBVR = 'N',  
     if SENSE = 'N', LWORK  $\geq$  max(1, 2  $\times$  N);  
     otherwise LWORK  $\geq$  max(1, N<sup>2</sup> + 6  $\times$  N).;  
 if JOBVL = 'V' or JOBVR = 'V',  
     if SENSE = 'N' or 'E', LWORK  $\geq$  max(1, 3  $\times$  N);  
     otherwise LWORK  $\geq$  max(1, N<sup>2</sup> + 6  $\times$  N)..

22: IWORK(\*) – INTEGER array *Workspace*

**Note:** the dimension of the array IWORK must be at least max(1, 2  $\times$  N – 1).

If SENSE = 'N' or 'E', IWORK is not referenced.

23: INFO – INTEGER *Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ , the  $QR$  algorithm failed to compute all the eigenvalues, and no eigenvectors or condition numbers have been computed; elements 1 : ILO – 1 and  $i + 1$  : N of WR and WI contain eigenvalues which have converged.

## 7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix  $(A + E)$ , where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and  $\epsilon$  is the *machine precision*. See Section 4.8 of Anderson *et al.* (1999) for further details.

## 8 Parallelism and Performance

F08NBF (DGEEVX) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08NBF (DGEEVX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

Each eigenvector is normalized to have Euclidean norm equal to unity and the element of largest absolute value real and positive.

The total number of floating-point operations is proportional to  $n^3$ .

The complex analogue of this routine is F08NPF (ZGEEVX).

## 10 Example

This example finds all the eigenvalues and right eigenvectors of the matrix

$$A = \begin{pmatrix} 0.35 & 0.45 & -0.14 & -0.17 \\ 0.09 & 0.07 & -0.54 & 0.35 \\ -0.44 & -0.33 & -0.03 & 0.17 \\ 0.25 & -0.32 & -0.13 & 0.11 \end{pmatrix},$$

together with estimates of the condition number and forward error bounds for each eigenvalue and eigenvector. The option to balance the matrix is used. In order to compute the condition numbers of the eigenvalues, the left eigenvectors also have to be computed, but they are not printed out in this example.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

### 10.1 Program Text

```

Program f08nbfe

!      F08NBF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: dgeevx, nag_wp, x02ajf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nb = 64, nin = 5, nout = 6
!      .. Local Scalars ..
Complex (Kind=nag_wp)      :: eig
Real (Kind=nag_wp)         :: abnrm, eps, tol
Integer                    :: i, ihi, ilo, info, j, lda, ldvl, &
                           ldvr, lwork, n
Logical                    :: pair
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:,,:), rconde(:), rcondv(:), &
                           scale(:), vl(:,,:), vr(:,,:), wi(:), &
                           work(:), wr(:)
Real (Kind=nag_wp)         :: dummy(1)
Integer, Allocatable       :: iwork(:)
!      .. Intrinsic Procedures ..
Intrinsic                  :: cmplx, max, nint
!      .. Executable Statements ..
Write (nout,*) 'F08NBF Example Program Results'
!      Skip heading in data file
Read (nin,*)
Read (nin,*) n
lda = n
ldvl = n
ldvr = n
lwork = (2+nb)*n
Allocate (a(lda,n),rconde(n),rcondv(n),scale(n),vl(ldvl,n),vr(ldvr,n), &
         wi(n),wr(n),iwork(2*n-2))

!      Use routine workspace query to get optimal workspace.
lwork = -1
!      The NAG name equivalent of dgeevx is f08nbf
Call dgeevx('Balance','Vectors (left)','Vectors (right)', &
         'Both reciprocal condition numbers',n,a,lda,wr,wi,vl,ldvl,vr,ldvr,ilo, &
         ihi,scale,abnrm,rconde,rcondv,dummy,lwork,iwork,info)

!      Make sure that there is enough workspace for blocksize nb.
lwork = max((nb+2)*n,nint(dummy(1)))
Allocate (work(lwork))

!      Read the matrix A from data file

```

```

      Read (nin,*)(a(i,1:n),i=1,n)

!      Solve the eigenvalue problem
!      The NAG name equivalent of dgeevx is f08nbf
      Call dgeevx('Balance','Vectors (left)','Vectors (right)', &
        'Both reciprocal condition numbers',n,a,lda,wr,wi,vl,ldvl,vr,ldvr,ilo, &
        ihi,scale,abnrm,rconde,rcondv,work,lwork,iwork,info)

      If (info==0) Then

!          Compute the machine precision
          eps = x02ajf()
          tol = eps*abnrm
          pair = .False.

!          Print the eigenvalues and vectors, and associated condition
!          number and bounds

          Write (nout,*)
          Write (nout,*) 'Eigenvalues'
          Write (nout,*)
          Write (nout,*) '          Eigenvalue          rcond    error'

          Do j = 1, n

!              Print information on jth eigenvalue

              If (wi(j)==0.0_nag_wp) Then
                If (rconde(j)>0.0_nag_wp) Then
                  If (tol/rconde(j)<10.0_nag_wp*eps) Then
                    Write (nout,99999) j, wr(j), rconde(j), '-'
                  Else
                    Write (nout,99998) j, wr(j), rconde(j), tol/rconde(j)
                  End If
                Else
                  Write (nout,99998) j, wr(j), rconde(j), 'Inf'
                End If
              Else
                If (rconde(j)>0.0_nag_wp) Then
                  If (tol/rconde(j)<10.0_nag_wp*eps) Then
                    Write (nout,99997) j, wr(j), wi(j), rconde(j), '-'
                  Else
                    Write (nout,99996) j, wr(j), wi(j), rconde(j), tol/rconde(j)
                  End If
                Else
                  Write (nout,99997) j, wr(j), wi(j), rconde(j), 'Inf'
                End If
              End If
            End Do

          Write (nout,*)
          Write (nout,*) 'Eigenvectors'
          Write (nout,*)
          Write (nout,*) '          Eigenvector          rcond    error'

          Do j = 1, n

!              Print information on jth eigenvector
              Write (nout,*)

!              Make real eigenvectors have positive first entry
              If (wi(j)==0.0E0_nag_wp) Then
                If (vr(1,j)<0.0_nag_wp) Then
                  vr(1:n,j) = -vr(1:n,j)
                End If
              If (rcondv(j)>0.0_nag_wp) Then
                If (tol/rcondv(j)<10.0_nag_wp*eps) Then
                  Write (nout,99999) j, vr(1,j), rcondv(j), '-'
                Else
                  Write (nout,99998) j, vr(1,j), rcondv(j), tol/rcondv(j)
                End If
              End If
            End Do
          End Do

```

```

      Else
        Write (nout,99998) j, vr(1,j), rcondv(j), 'Inf'
      End If
      Write (nout,99995) vr(2:n,j)
    Else
      If (pair) Then
        eig = cmplx(vr(1,j-1),-vr(1,j),kind=nag_wp)
      Else
        eig = cmplx(vr(1,j),vr(1,j+1),kind=nag_wp)
      End If
      If (rcondv(j)>0.0_nag_wp) Then
        If (tol/rcondv(j)<10.0_nag_wp*eps) Then
          Write (nout,99997) j, eig, rcondv(j), '-'
        Else
          Write (nout,99996) j, eig, rcondv(j), tol/rcondv(j)
        End If
      Else
        Write (nout,99997) j, eig, rcondv(j), 'Inf'
      End If
      If (pair) Then
        Write (nout,99994)(vr(i,j-1),-vr(i,j),i=2,n)
      Else
        Write (nout,99994)(vr(i,j),vr(i,j+1),i=2,n)
      End If
      pair = .Not. pair
    End If
  End Do
  Write (nout,*)
  Write (nout,*) 'Errors below 10*machine precision are not displayed'
Else
  Write (nout,*)
  Write (nout,99993) 'Failure in DGEEVX. INFO = ', info
End If

99999 Format (1X,I2,2X,1P,E11.4,14X,0P,F7.4,4X,A)
99998 Format (1X,I2,2X,1P,E11.4,11X,0P,F7.4,1X,1P,E8.1)
99997 Format (1X,I2,1X,'( ',1P,E11.4,' ',',',E11.4,')',1X,0P,F7.4,4X,A)
99996 Format (1X,I2,1X,'( ',1P,E11.4,' ',',',E11.4,')',1X,0P,F7.4,1X,1P,E8.1)
99995 Format (1X,4X,1P,E11.4)
99994 Format (1X,3X,'( ',1P,E11.4,' ',',',E11.4,')')
99993 Format (1X,A,I4)
      End Program f08nbfe

```

## 10.2 Program Data

```

F08NBF Example Program Data
  4                               :Value of N
  0.35   0.45  -0.14  -0.17
  0.09   0.07  -0.54   0.35
 -0.44  -0.33  -0.03   0.17
  0.25  -0.32  -0.13   0.11 :End of matrix A

```

## 10.3 Program Results

F08NBF Example Program Results

Eigenvalues

	Eigenvalue	rcond	error
1	7.9948E-01	0.9936	-
2	(-9.9412E-02, 4.0079E-01)	0.7027	-
3	(-9.9412E-02, -4.0079E-01)	0.7027	-
4	-1.0066E-01	0.5710	-

Eigenvectors

	Eigenvector	rcond	error
1	6.5509E-01 5.2363E-01	0.6252	-



```
-5.3622E-01
 9.5607E-02

2 (-1.9330E-01, 2.5463E-01) 0.3996 -
 ( 2.5186E-01,-5.2240E-01)
 ( 9.7182E-02,-3.0838E-01)
 ( 6.7595E-01, 0.0000E+00)

3 (-1.9330E-01,-2.5463E-01) 0.3996 -
 ( 2.5186E-01, 5.2240E-01)
 ( 9.7182E-02, 3.0838E-01)
 ( 6.7595E-01,-0.0000E+00)

4 1.2533E-01          0.3125 -
 3.3202E-01
 5.9384E-01
 7.2209E-01
```

Errors below 10\*machine precision are not displayed

---